
Overview 

The BMP085 is a basic sensor that is designed specifically for measuring barometric

pressure (it also does temperature measurement on the side to help). It's one of the

few sensors that does this measurement, and its fairly low cost so you'll see it used a

lot. You may be wondering why someone would want to measure atmospheric

pressure, but its actually really useful for two things. One is to measure altitude. As

we travel from below sea level to a high mountain, the air pressure decreases. That

means that if we measure the pressure we can determine our altitude - handy when

we don't want the expense or size of a GPS unit. Secondly, atmospheric pressure can

be used as a predictor of weather which is why weather-casters often talk about

"pressure systems" 

Specifications

Pressure sensing range: 300-1100 hPa (9000m to -500m above sea level)

Up to 0.03hPa / 0.25m resolution

-40 to +85°C operational range, +-2°C temperature accuracy

2-pin i2c interface on chip

V1 of the breakout uses 3.3V power and logic level only

V2 of the breakout uses 3.3-5V power and logic level for more flexible usage

The BMP085 has been discontinued by BOSCH and replaced with the BMP180 

which is smaller and less expensive but is identical in terms of wiring and 

software! 

 

• 

• 

• 

• 

• 

• 

Datasheet () 

You can pick up a BMP085 breakout board from the Adafruit shop! (http://adafru.it/

391)

Wiring the BMP 

Since the BMP085 is a i2c sensor, its very easy to wire up. We'll be using an Arduino

as an example but any microcontroller with i2c can be used. To start using with a

solderless breadboard, we need to solder the header pins onto the breakout board.

We suggest plugging the header into a breadboard so the long pins are in the

breadboard, and then laying the BMP085 breakout board on top. The photos show a

v1 BMP085 sensor but this part is identical for both versions!

Then solder all of the pins! 

• 

That's it! Now we can wire the board up to the microcontroller. 

Connect the VCC pin to a 3.3V power source. The V1 of the sensor breakout cannot

be used with anything higher than 3.3V so don't use a 5V supply! V2 of the sensor

board has a 3.3V regulator so you can connect it to either 3.3V or 5V if you do not

have 3V available. 

Connect GND to the ground pin.

Connect the i2c SCL clock pin to your i2c clock pin. On the classic Arduino Uno/

Duemilanove/Diecimila/etc this is Analog pin #5

Connect the i2c SDA data pin to your i2c data pin. On the classic Arduino Uno/

Duemilanove/Diecimila/etc this is Analog pin #4

Unfortunately, the i2c lines on most microcontrollers are fixed so you're going to have

to stick with those pins.

You don't need to connect the XCLR (reset) or EOC (end-of-conversion) pins. If you

need to speed up your conversion time, you can use the EOC as a indicator - in our

code we just hang out and wait the maximum time possible.

Using the BMP (API v2) 

This page is based on the new v2 of the BMP085 driver, which uses Adafruit's new Un

ified Sensor Driver (). The driver provides better support for altitude calculations, and

makes it easy to switch between the BMP085 and any other supported pressure

sensor in your projects.

If you haven't already done so, you'll need to install the Adafrut_Sensor library () on

your system as well, since Adafruit_BMP085 relies on this library to generate the

sensor data in a universal manner.

Using the BMP085 / BMP180

To use this sensor and calculate the altitude and barometric pressure, there's a lot of

very hairy and unpleasant math. You can check out the math in the datasheet but

really, its not intuitive or educational - its just how the sensor works. So we took care

of all the icky math and wrapped it up into a nice Arduino library. 

You can find the Arduino library repository on github () To install it, click this button to

download the compressed ZIP file then install it. This guide () will help you with the

install process if you have never installed an Arduino library.

Download the Adafruit_BMP085

Arduino Library (API v2)

The same code is used for both the BMP085 and BMP180 (they are compatible!)

Restart the IDE

Now you can run this first example sketch

#include &lt;Wire.h&gt;
#include &lt;Adafruit_Sensor.h&gt;
#include &lt;Adafruit_BMP085_U.h&gt;
   
Adafruit_BMP085_Unified bmp = Adafruit_BMP085_Unified(10085);

void setup(void) 
{
  Serial.begin(9600);
  Serial.println("Pressure Sensor Test"); Serial.println("");
  
  /* Initialise the sensor */
  if(!bmp.begin())
  {
    /* There was a problem detecting the BMP085 ... check your connections */
    Serial.print("Ooops, no BMP085 detected ... Check your wiring or I2C ADDR!");
    while(1);
  }
}

void loop(void) 
{
  /* Get a new sensor event */ 
  sensors_event_t event;
  bmp.getEvent(&amp;event);
 
  /* Display the results (barometric pressure is measure in hPa) */
  if (event.pressure)
  {

    /* Display atmospheric pressure in hPa */
    Serial.print("Pressure: "); Serial.print(event.pressure); Serial.println(" 
hPa");
  }
  else
  {
    Serial.println("Sensor error");
  }
  delay(250);
}

Then open up the serial monitor at 9600 baud. The sketch will continuously print out

the pressure in hPa (hectoPascals). You can test that the sensor is measuring

variations in pressure by placing your fingertip over the open port hole in the top of

the sensor. The pressure will increase as you can see here: 

Altitude Measurements

Since we know that pressure drops as we gain altitude (that's why air is so thin on

mountain-tops) we can compute the current altitude knowing the pressure and

temperature. Again, there's a bit of hairy math involved, you can read about the

calculations on wikipedia (where this graph is from) (). 

With the Arduino library, we take care of that for you! Simply update the 'void loop()'

function above with the code below to get the altitude based on the pressure and

temperature: 

void loop(void) 
{
  /* Get a new sensor event */ 
  sensors_event_t event;
  bmp.getEvent(&amp;event);
 
  /* Display the results (barometric pressure is measure in hPa) */
  if (event.pressure)
  {
    /* Display atmospheric pressue in hPa */
    Serial.print("Pressure:    ");
    Serial.print(event.pressure);
    Serial.println(" hPa");
    
    /* Calculating altitude with reasonable accuracy requires pressure    *
     * sea level pressure for your position at the moment the data is     *
     * converted, as well as the ambient temperature in degress           *
     * celcius.  If you don't have these values, a 'generic' value of     *
     * 1013.25 hPa can be used (defined as SENSORS_PRESSURE_SEALEVELHPA   *
     * in sensors.h), but this isn't ideal and will give variable         *
     * results from one day to the next.                                  *
     *                                                                    *
     * You can usually find the current SLP value by looking at weather   *
     * websites or from environmental information centers near any major  *
     * airport.                                                           *
     *                                                                    *
     * For example, for Paris, France you can check the current mean      *
     * pressure and sea level at: http://bit.ly/16Au8ol                   */
     
    /* First we get the current temperature from the BMP085 */
    float temperature;
    bmp.getTemperature(&amp;temperature);
    Serial.print("Temperature: ");
    Serial.print(temperature);
    Serial.println(" C");

    /* Then convert the atmospheric pressure, SLP and temp to altitude    */
    /* Update this next line with the current SLP for better results      */
    float seaLevelPressure = SENSORS_PRESSURE_SEALEVELHPA;
    Serial.print("Altitude:    "); 
    Serial.print(bmp.pressureToAltitude(seaLevelPressure,
                                        event.pressure,
                                        temperature)); 
    Serial.println(" m");
    Serial.println("");
  }
  else
  {
    Serial.println("Sensor error");
  }
  delay(1000);
}

Run the sketch to see the calculated altitude. 

The data above is reasonably close to what I'd expect at my location, but we can

improve the accuracy by changing the reference sea level pressure, which will

change depending on the weather conditions. Every 1 hPa that we are off on the sea

level pressure equals about 8.5 m of error in the altitude calculations!

Many weather sites, (particularly near major airports) will provide pressure readings. If

you happened to be near Paris, France, for example, you might look up the current air

pressure at Charles de Gaulle airport (), which we can see is 1009 hPa (a meaningful

difference from the generoc 1013.25 hPa value we are plugging in via the SENSORS_

PRESSURE_SEALEVELHPA macro): 

Updating the following line to 1009 will give us a more accurate altitude: 

float seaLevelPressure = 1009;

This now gives us the following results, which shows that calibrating for your local

conditions is often worthwhile when working with low altitudes! 

Just be careful looking for local mean pressure at sea level values, since the functions

in the driver are expecting hPa units, not one of the dozens of other values you may

encounter, but you should be able to convert anything you find to hPa () which is a

standard SI unit. 

 

If you don't know the current mean pressure at sea level for your current weather 

conditions and location, SENSORS_PRESSURE_SEALEVELHPA is still a safe 

starting point: 

Using the BMP (API v1) 

To use this sensor and calculate the altitude and barometric pressure, there's a lot of

very hairy and unpleasant math. You can check out the math in the datasheet but

really, its not intuitive or educational - its just how the sensor works. So we took care

of all the icky math and wrapped it up into a nice Arduino library. 

You can find the Arduino library repository on github () To install it, click this button to

download the compressed ZIP file then install it. This guide () will help you with the

install process if you have never installed an Arduino library.

Download the Adafruit_BMP085

Arduino Library (API v1)

The same driver is used for both the BMP085 and BMP180!

Restart the IDE

Now you can run this first example sketch

#include "Wire.h"
#include "Adafruit_BMP085.h"
 
Adafruit_BMP085 bmp;
 
void setup() {
  Serial.begin(9600);
  bmp.begin();  
}
 
void loop() {
    Serial.print("Temperature = ");
    Serial.print(bmp.readTemperature());
    Serial.println(" *C");
 
    Serial.print("Pressure = ");
    Serial.print(bmp.readPressure());
    Serial.println(" Pa");
 
    Serial.println();
    delay(500);
}

Then open up the serial monitor at 9600 baud. The sketch will continuously print out

the temperature in °C and pressure in Pa (Pascals). You can test that the sensor is

This page applies to the old sensor driver, and is provided here for reference 

sake. We recommend using the new 'Unified' BMP085 driver (API v2) for any 

future projects. 

measuring variations in temperature and pressure by placing your fingertip over the

open port hole in the top of the sensor. The temperature and pressure will increase as

you can see here: 

Altitude Measurements

Since we know that pressure drops as we gain altitude (that's why air is so thin on

mountain-tops) we can compute the current altitude knowing the pressure and

temperature. Again, there's a bit of hairy math involved, you can read about the

calculations on wikipedia (where this graph is from) (). 

With the Arduino library, we take care of that for you! Simply run this sketch which will

return the current altitude based on the pressure. 

#include "Wire.h"
#include "Adafruit_BMP085.h"
 
Adafruit_BMP085 bmp;
 
void setup() {
  Serial.begin(9600);
  bmp.begin();  
}
 
void loop() {
    Serial.print("Temperature = ");
    Serial.print(bmp.readTemperature());
    Serial.println(" *C");
 
    Serial.print("Pressure = ");
    Serial.print(bmp.readPressure());
    Serial.println(" Pa");
 
    // Calculate altitude assuming 'standard' barometric
    // pressure of 1013.25 millibar = 101325 Pascal
    Serial.print("Altitude = ");
    Serial.print(bmp.readAltitude());
    Serial.println(" meters");
 
    Serial.println();
    delay(500);
}

Run the sketch to see the calculated altitude. 

For example, according to the sensor we are 21.5m below sea level. Only problem is, I

know for a fact that our current location is not below sea level! So what's wrong with

the sensor? Turns out the sensor is just fine. The problem is that the pressure at sea

level changes with the weather. So we need to 'normalize' the sensor, and let it know

what the sea-level pressure is. You can look up the current sea level pressure on any

weather site. 

Unfortunately there are half-dozen different units of pressure. here we see it in

inches, that's technically "Mercury Inches" or "Hg Inches We need it in Pascals, so

we'll convert it! () 

OK so that's 101,964 Pascals. Open up the Examples->BMP085test example from the

Arduino IDE menubar and edit the line where you pass in the 'corrected' altitude. 

Now it will print out the correct altitude! 30 meters which is a lot better. 

Downloads 

Files

Fritzing objects available on GitHub () 

EagleCAD PCBs for the BMP085 on GitHub () 

EagleCAD PCBs for the BMP180 on GitHub () 

BMP180 Datasheet () 

 

 

 

• 

• 

• 

• 

BMP085 Datasheet () 

Firmware

Unified BMP085 Driver Repository (New)

Github repository for the 'Unified' sensor driver, inlcuding example code ()

For new projects, it's highly recommended to use the new Unified BMP085 Driver,

which makes it easier to switch between other pressure sensors, and work with

standard SI units of measure.

To download, click the DOWNLOAD button in the top right corner, renamed the

compressed folder Adafruit_BMP085. Check that the Adafruit_BMP085 folder

contains Adafruit_BMP085.cpp and Adafruit_BMP085.h.

Place the Adafruit_BMP085 library folder in your /libraries folder. You may need to

create the libraries subfolder if its your first library. Restart the IDE.

Original BMP085 Repository (Deprecated)

Github repository with library and code examples

 ()

To download click the DOWNLOADS button in the top right corner, rename the

uncompressed folder Adafruit_BMP085. Check that the Adafruit_BMP085 folder

contains Adafruit_BMP085.cpp and Adafruit_BMP085.h

Place the Adafruit_BMP085 library folder your /libraries/ folder. You may need to

create the libraries subfolder if its your first library. Restart the IDE.

• 

This repository is not recommended for new projects, but is provided for 

reference sake for projects that made use of the BMP085 before the updated 

Unified driver was published 

